Math 2650 - Linear Differential Equations

A. J. Meir

Example: Pendulum

An ordinary differential equation is an equation which involves an unknown function and its derivatives with respect to one independent variable. (A partial differential equation involves an unknown function and its partial derivatives with respect to several independent variables.)

The differential equation (initial value problem) describing a pendulum

$$m \frac{d^2}{dt^2}\phi + \frac{c}{l}\frac{d}{dt}\phi + \frac{m\,g}{l}\sin(\phi) = \mathrm{F(t)} \qquad \phi(0) = \phi_0 \qquad \frac{d}{dt}\phi(0) = \phi_1$$

we usually approximate this nonlinear equation, by the linear equation

$$m \frac{d^2}{dt^2}\phi + \frac{c}{l}\frac{d}{dt}\phi + \frac{m\,g}{l}\phi = \mathrm{F(t)} \qquad \phi(0) = \phi_0 \qquad \frac{d}{dt}\phi(0) = \phi_1$$

which is a good approximation when the angle ϕ is small. Here m is the mass of the bob, c the amount of damping, l length of pendulum, and g acceleration of gravity. The initial condition ϕ_0 is the initial angle and ϕ_1 the initial velocity. Below we assume $F(t) = 0$.

Maple commends

```
> restart:with(plottools):with(plots):
Warning, the name changecoords has been redefined
Warning, the previous binding of the name arrow has been removed and it now has an
assigned value
```

```
> pendulum:=proc(m,c,l,g,T,phi0,phi1)
> local eq, ic, soln, p, L, C, N, seq1, seq2, h, support, bob:
> global n, x, y, graph:
> eq:=m*diff(phi(t),t,t)+(c/l)*diff(phi(t),t)+(m*g/l)*phi(t)=0:
> ic:=phi(0)=phi0,D(phi)(0)=phi1:
> soln:=rhs(dsolve({eq,ic},phi(t))):
> graph:=plot({soln,diff(soln,t)},t=0..T,thickness=2):
> p:=unapply(soln,t):
```

```
> L:=(x,y)->line([0,0],[x,y],thickness=3,color=red):
> C:=(x,y)->disk([x,y],0.3,color=blue):
> N:=200:
> x:=l*sin(p((T/N)*n)):y:=-l*cos(p((T/N)*n)):
> seq1:=seq(L(x,y),n=0..N):
> seq2:=seq(C(x,y),n=0..N):
> h:=line([-4,0],[4,0],thickness=5,color=green):
> support:=display(seq1,insequence=true):
> bob:=display(seq2,insequence=true):
> display(h,support,bob,scaling=constrained):
> end:
> Mass:=1/20:Damping:=0.03:Length:=4:G:=10:Time:=20:
  InitAngle:=1:InitVelocity:=-1:
> pendulum(Mass,Damping,Length,G,Time,InitAngle,InitVelocity);graph;
```




1.5
1
0.5
0
-0.5
-1
-1.5
t
0
5
10
15
20

>

>